Multi-headed Lattice Green Function (N = 4, M = 3)

```
In[ ]:= NN = 4;
MM = 3;
```

Recall some basic definitions in the paper:

$$P_{M,N}(z) := \frac{1}{(2\pi)^N} \int_{-\pi}^{\pi} \cdots \int_{-\pi}^{\pi} \frac{1}{1 - \frac{z}{\binom{N}{M}} \sigma_M(\cos\theta_1, \ldots, \cos\theta_N)} \, d\theta_1 \ldots d\theta_N$$

$$R_{M,N}(z) := P_{M,N}\left(2^M \binom{N}{M} z\right) \text{ and } R_{M,N}(z) = \sum_{n \geq 0} r_{M,N}(n)\, z^n$$

Also, for M odd or $M = N$, we always have $r(2\,n+1) = 0$. Hence, define
$\tilde{r}_{M,N}(n) := r_{M,N}(2\,n)$ and $\tilde{R}_{M,N}(z) := \sum_{n \geq 0} \tilde{r}_{M,N}(n)\, z^n = \sum_{n \geq 0} r_{M,N}(2\,n)\, z^n$

Our goal is to find:

Case 1. *M* even and *M* ≠ *N*:

- recurrences (REC) for $r(n)$ or differential equations (ODE) for $R(z)$.

Case 2. *M* odd or *M* = *N*:

- recurrences (REC) for $\tilde{r}(n)$ or differential equations (ODE) for $\tilde{R}(z)$.

Command: UnrollRecurrence

Generate a sequence from recurrence & initial values (Koutschan’s implementation).

```
In[ ]:= (* given a recurrence rec in f[n], compute the values {f[0],f[1],...,f[bound]}
 where inits are the initial values
 {f[0],...,f[d-1]} with d being the order of the recurrence *)
Clear[UnrollRecurrence];
UnrollRecurrence[rec1_, f_[n_], inits_, bound_] :=
  Module[{i, x, vals = inits, rec = rec1},
   If[Head[rec] =!= Equal, rec = (rec == 0)];
   rec = rec /. n → n - Max[Cases[rec, f[n + a_.] :→ a, Infinity]];
   Do[
    AppendTo[vals, Solve[rec /. n → i /. f[i] → x /. f[a_] :→ vals[[a + 1]], x][[1, 1, 2]]];
    , {i, Length[inits], bound}];
   Return[vals];
  ];
```

Load RISC packages.

```
In[]:= << RISC`HolonomicFunctions`
       << RISC`Asymptotics`
       << RISC`Guess`
```

HolonomicFunctions Package version 1.7.3 (21-Mar-2017)
written by Christoph Koutschan


```
--> Type  ?HolonomicFunctions  for help.
```

Asymptotics Package version 0.3
written by Manuel Kauers

Package GeneratingFunctions version 0.9 written by Christian Mallinger

Guess Package version 0.52
written by Manuel Kauers

Apply creative telescoping to $\tilde{r}(n)$.

```
In[]:= ClearAll[k1, k2, k3, k4, z, w, α, β];

In[]:= k4 = α - k1 - k2 - k3;
       summand = Binomial[2 α, 2 k1] Binomial[2 α - 2 k1, 2 k2]
          Binomial[2 α - 2 k1 - 2 k2, 2 k3] Binomial[2 (α - k1), α - k1]
          Binomial[2 (α - k2), α - k2] Binomial[2 (α - k3), α - k3] Binomial[2 (α - k4), α - k4];

In[]:= Timing[ann0 = Annihilator[summand, {S[k1], S[k2], S[k3], S[α]}];]
Out[]= {0.046875, Null}

In[]:= Timing[ann1 = FindCreativeTelescoping[ann0, S[k1] - 1][[1]];]
Out[]= {37.2969, Null}

In[]:= Timing[ann2 = FindCreativeTelescoping[ann1, S[k2] - 1][[1]];]
Out[]= {347.047, Null}

In[]:= Timing[ann3 = FindCreativeTelescoping[ann2, S[k3] - 1][[1]];]
Out[]= {291.984, Null}
```

Alternatively, you may import the value of ann3 from an external file.

```
In[•]:= ann3 = ToExpression[Import[NotebookDirectory[] <> "Data-N4M3-Sum.txt"]];
```

ann3 gives a REC for $\tilde{r}(n)$.

Compute the REC for $\tilde{r}(n)$.

Order 4

```
In[•]:= RECNormalizedinS = NormalizeCoefficients[ann3[[1]]];

In[•]:= RECNormalizedinSOrder = OrePolynomialDegree[RECNormalizedinS, S[α]]

Out[•]= 4
```

We also write this REC explicitly.

SeqNormalized gives the REC in Theorem 4.3! (To be displayed at the end of this notebook)

```
In[•]:= ClearAll[Seq];
SeqNormalized = ApplyOreOperator[RECNormalizedinS, Seq[α]];
```

The initial values of $\tilde{r}(n)$ are as follows.

```
In[•]:= SeqListIni = {};

MAX = 20;

For[n = 0, n ≤ MAX, n++,
  coord = Select[Tuples[Table[i, {i, 0, n}], NN], Total[#] == n &];
  size = Length@coord;
  p = Sum[Multinomial[Sequence @@ (2 coord[[i]])] *
     Product[Binomial[2 n - 2 coord[[i, j]], n - coord[[i, j]]], {j, 1, NN}], {i, 1, size}];
  SeqListIni = Append[SeqListIni, p];
 ];

SeqListIni

seq[n_] := SeqListIni[[n + 1]];

Out[•]= {1, 32, 6048, 2 451 200, 1 391 236 000, 921 422 380 032, 663 895 856 219 904, 505 041 413 866 868 736,
 399 445 932 990 555 902 880, 325 440 143 503 901 735 429 120, 271 445 584 301 606 582 663 031 808,
 230 773 066 339 125 955 854 130 661 376, 199 326 200 240 673 646 611 787 771 995 904,
 174 478 237 021 099 598 812 491 315 604 889 600, 154 480 035 620 813 053 446 642 174 412 128 768 000,
 138 129 336 609 134 098 952 004 475 839 318 761 472 000,
 124 577 089 053 969 968 356 059 653 140 361 638 344 938 400,
 113 209 463 052 287 193 655 237 025 876 331 530 870 707 737 600,
 103 573 496 015 054 055 969 039 980 718 499 533 706 000 571 520 000,
 95 328 837 240 197 678 160 114 853 748 204 677 385 026 223 109 120 000,
 88 215 610 025 056 975 283 519 690 346 309 846 200 279 286 296 474 496 000}
```

Now we may numerically verify our REC.

```
In[•]:= Table[SeqNormalized /. {Seq → seq, α → n}, {n, 0, MAX - RECNormalizedinSOrder}]

Out[•]= {0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0, 0}
```

Let us the generate a list of $\tilde{r}(n)$.

```
In[∘]:= Bound = 5000;

SeqList = UnrollRecurrence[SeqNormalized, Seq[α], SeqListIni, Bound];

seq[n_] := SeqList[[n + 1]];
```

Get the ODE for $\tilde{R}(z)$.

ODENormalizedinD - in terms of the derivation operator *D*

ODENormalizedinTheta - in terms of the derivation operator θ - **Order 24, Degree 4**

```
In[∘]:= ODENormalizedinD = NormalizeCoefficients[DFiniteRE2DE[{RECNormalizedinS}, {α}, {w}][[1]]];

In[∘]:= ODENormalizedinTheta =
          NormalizeCoefficients[ChangeOreAlgebra[w ** ODENormalizedinD, OreAlgebra[Euler[w]]]];

In[∘]:= ODENormalizedinThetaOrder = OrePolynomialDegree[ODENormalizedinTheta, Euler[w]]

Out[∘]= 24

In[∘]:= ODENormalizedinThetaDegree =
         Max[Exponent[OrePolynomialListCoefficients[ODENormalizedinTheta], w]]

Out[∘]= 4
```

Get the ODE for $R(z)$.

ODEinD - in terms of the derivation operator *D*

ODEinTheta - in terms of the derivation operator θ - **Order 24, Degree 8 (Refer to Table 1)**

```
In[∘]:= ODEinD = NormalizeCoefficients[
           DFiniteSubstitute[{ODENormalizedinD}, {w → z^2}, Algebra → OreAlgebra[Der[z]]][[1]]];

In[∘]:= ODEinTheta = NormalizeCoefficients[ChangeOreAlgebra[z ** ODEinD, OreAlgebra[Euler[z]]]];

In[∘]:= ODEinThetaOrder = OrePolynomialDegree[ODEinTheta, Euler[z]]

Out[∘]= 24

In[∘]:= ODEinThetaDegree = Max[Exponent[OrePolynomialListCoefficients[ODEinTheta], z]]

Out[∘]= 8
```

Guess a Minimal REC for $\tilde{r}(n)$.

Its order is 4, and is identical to that of the REC in Theorem 4.3 (RECNormalizedinS).

```
In[∘]:= ClearAll[Seq];
        RECGuess = GuessMinRE[Take[SeqList, 300], Seq[α]];
        RECGuessinS = NormalizeCoefficients[ToOrePolynomial[RECGuess /. {Seq[k_] → S[α]^(k-α)}]];

In[∘]:= RECGuessinSOrder = OrePolynomialDegree[RECGuessinS, S[α]]

Out[∘]= 4
```

Compute the asymptotics for $\tilde{r}(n)$.

```
In[•]:= AsyList = Asymptotics[SeqNormalized, Seq[α]];
        N[AsyList]
```

Out[•]= $\left\{\frac{16.^{\alpha}}{\alpha^2}, \frac{256.^{\alpha}}{\alpha^2}, \frac{1024.^{\alpha}}{\alpha^3}, \frac{1024.^{\alpha}}{\alpha^2}\right\}$

```
In[•]:= Ind = Reverse[Table[Floor[Bound / i], {i, 1, 3}]]
        Table[N[seq[Ind[[i]]] / (AsyList[[1]] /. {α → Ind[[i]]})], {i, 1, Length@Ind}]
        Table[N[seq[Ind[[i]]] / (AsyList[[2]] /. {α → Ind[[i]]})], {i, 1, Length@Ind}]
        Table[N[seq[Ind[[i]]] / (AsyList[[3]] /. {α → Ind[[i]]})], {i, 1, Length@Ind}]
        Table[N[seq[Ind[[i]]] / (AsyList[[4]] /. {α → Ind[[i]]})], {i, 1, Length@Ind}]
```

Out[•]= {1666, 2500, 5000}

Out[•]= $\{2.806687457612096 \times 10^{3007}, 6.343600724639624 \times 10^{4513}, 1.787780641892824 \times 10^{9029}\}$

Out[•]= $\{2.422718463768892 \times 10^{1001}, 3.179649140402995 \times 10^{1503}, 4.491598734476526 \times 10^{3008}\}$

Out[•]= {37.5001, 56.2783, 112.568}

Out[•]= {0.0225091, 0.0225113, 0.0225136}

Approximate the Polya number.

In[•]:= AtOne = N[Sum[seq[n] * $\left(\frac{1}{2^{MM}\,\text{Binomial[NN, MM]}}\right)^{2n}$, {n, 0, Bound}], 11]

N[1 - $\frac{1}{\text{AtOne}}$, 10]

Out[•]= 1.0452834156

Out[•]= 0.04332166274

Display the REC for $\tilde{r}(n)$ in Theorem 4.3

```
In[•]:= Collect[Expand[-SeqNormalized], Seq[_]]
```

Out[•]=

$$
\begin{aligned}
&\big(221\,086\,792\,032\,258\,663\,383\,040 + 3\,002\,581\,182\,281\,579\,476\,549\,632\,\alpha + \\
&\quad 18\,896\,284\,453\,973\,181\,469\,818\,880\,\alpha^{2} + 73\,337\,056\,136\,834\,742\,984\,114\,176\,\alpha^{3} + \\
&\quad 197\,017\,275\,538\,043\,925\,583\,364\,096\,\alpha^{4} + 389\,745\,626\,428\,476\,129\,286\,291\,456\,\alpha^{5} + \\
&\quad 589\,529\,476\,016\,351\,811\,509\,157\,888\,\alpha^{6} + 698\,690\,177\,713\,813\,455\,561\,031\,680\,\alpha^{7} + \\
&\quad 659\,396\,154\,092\,196\,671\,988\,432\,896\,\alpha^{8} + 500\,766\,687\,956\,261\,350\,615\,810\,048\,\alpha^{9} + \\
&\quad 307\,887\,490\,552\,535\,839\,569\,608\,704\,\alpha^{10} + 153\,616\,793\,330\,862\,792\,246\,296\,576\,\alpha^{11} + \\
&\quad 62\,125\,104\,506\,185\,984\,379\,977\,728\,\alpha^{12} + 20\,265\,270\,278\,609\,884\,774\,662\,144\,\alpha^{13} + \\
&\quad 5\,282\,843\,409\,745\,454\,510\,899\,200\,\alpha^{14} + 1\,084\,193\,901\,809\,507\,676\,192\,768\,\alpha^{15} + \\
&\quad 171\,154\,981\,038\,855\,165\,050\,880\,\alpha^{16} + 20\,040\,031\,539\,432\,857\,272\,320\,\alpha^{17} + \\
&\quad 1\,638\,003\,152\,561\,664\,688\,128\,\alpha^{18} + 83\,373\,097\,696\,100\,352\,000\,\alpha^{19} + 1\,988\,330\,027\,074\,191\,360\,\alpha^{20}\big) \\
&\text{Seq}[\alpha] + \big(-123\,596\,648\,884\,357\,621\,088\,256 - 1\,387\,410\,081\,329\,207\,115\,251\,712\,\alpha - \\
&\quad 7\,308\,010\,505\,383\,031\,273\,947\,136\,\alpha^{2} - 24\,020\,604\,752\,075\,269\,740\,691\,456\,\alpha^{3} - \\
&\quad 55\,262\,591\,055\,735\,725\,773\,815\,808\,\alpha^{4} - 94\,607\,549\,345\,038\,165\,436\,006\,400\,\alpha^{5} - \\
&\quad 125\,070\,786\,847\,359\,746\,869\,821\,440\,\alpha^{6} - 130\,760\,992\,638\,503\,780\,446\,109\,696\,\alpha^{7} - \\
&\quad 109\,819\,712\,522\,499\,293\,630\,693\,376\,\alpha^{8} - 74\,830\,049\,897\,678\,615\,099\,736\,064\,\alpha^{9} - \\
&\quad 41\,599\,115\,200\,046\,517\,939\,601\,408\,\alpha^{10} - 18\,902\,277\,196\,351\,684\,209\,803\,264\,\alpha^{11} - \\
&\quad 7\,008\,965\,526\,989\,775\,347\,122\,176\,\alpha^{12} - 2\,109\,519\,207\,312\,665\,281\,560\,576\,\alpha^{13} - \\
&\quad 510\,375\,764\,108\,304\,797\,663\,232\,\alpha^{14} - 97\,744\,104\,267\,386\,959\,429\,632\,\alpha^{15} - \\
&\quad 14\,472\,279\,363\,085\,494\,386\,688\,\alpha^{16} - 1\,596\,811\,738\,769\,963\,089\,920\,\alpha^{17} - 123\,530\,156\,260\,699\,668\,480 \\
&\quad\; \alpha^{18} - 5\,975\,058\,303\,292\,538\,880\,\alpha^{19} - 135\,920\,997\,944\,524\,800\,\alpha^{20}\big)\ \text{Seq}[1+\alpha] + \\
&\big(2\,413\,729\,498\,666\,800\,513\,024 + 25\,435\,086\,835\,865\,925\,058\,560\,\alpha + 125\,542\,481\,225\,411\,227\,975\,680\,\alpha^{2} + \\
&\quad 386\,097\,946\,352\,750\,392\,590\,336\,\alpha^{3} + 830\,183\,396\,028\,360\,968\,208\,384\,\alpha^{4} + \\
&\quad 1\,327\,255\,653\,860\,270\,011\,465\,728\,\alpha^{5} + 1\,637\,850\,112\,836\,596\,110\,688\,256\,\alpha^{6} + \\
&\quad 1\,598\,197\,760\,043\,557\,807\,628\,288\,\alpha^{7} + 1\,252\,980\,911\,862\,994\,173\,739\,008\,\alpha^{8} + \\
&\quad 797\,358\,770\,338\,813\,407\,952\,896\,\alpha^{9} + 414\,276\,959\,391\,975\,941\,603\,328\,\alpha^{10} + \\
&\quad 176\,103\,421\,096\,866\,815\,410\,176\,\alpha^{11} + 61\,159\,515\,859\,482\,838\,548\,480\,\alpha^{12} + \\
&\quad 17\,263\,930\,413\,062\,410\,149\,888\,\alpha^{13} + 3\,923\,295\,133\,237\,310\,914\,560\,\alpha^{14} + \\
&\quad 706\,924\,713\,366\,338\,125\,824\,\alpha^{15} + 98\,652\,029\,401\,005\,981\,696\,\alpha^{16} + 10\,278\,087\,291\,823\,325\,184\,\alpha^{17} + \\
&\quad 752\,234\,327\,699\,226\,624\,\alpha^{18} + 34\,490\,272\,274\,841\,600\,\alpha^{19} + 745\,214\,176\,788\,480\,\alpha^{20}\big)\ \text{Seq}[2+\alpha] + \\
&\big(-9\,569\,617\,440\,812\,835\,840 - 97\,443\,791\,378\,162\,009\,856\,\alpha - 463\,583\,339\,186\,644\,316\,800\,\alpha^{2} - \\
&\quad 1\,370\,837\,922\,368\,778\,354\,176\,\alpha^{3} - 2\,827\,452\,328\,200\,593\,850\,560\,\alpha^{4} - 4\,326\,575\,055\,112\,730\,856\,640\,\alpha^{5} - \\
&\quad 5\,099\,519\,612\,920\,329\,528\,000\,\alpha^{6} - 4\,743\,666\,552\,937\,883\,189\,952\,\alpha^{7} - 3\,539\,068\,890\,050\,114\,722\,112\,\alpha^{8} - \\
&\quad 2\,139\,750\,587\,880\,300\,657\,856\,\alpha^{9} - 1\,054\,730\,779\,373\,468\,537\,920\,\alpha^{10} - 424\,824\,967\,934\,147\,228\,480\,\alpha^{11} - \\
&\quad 139\,643\,546\,214\,642\,867\,648\,\alpha^{12} - 37\,274\,084\,807\,088\,072\,384\,\alpha^{13} - 8\,003\,802\,897\,605\,020\,608\,\alpha^{14} - \\
&\quad 1\,361\,866\,764\,260\,304\,576\,\alpha^{15} - 179\,386\,646\,751\,384\,192\,\alpha^{16} - 17\,635\,678\,788\,631\,680\,\alpha^{17} - \\
&\quad 1\,217\,772\,669\,657\,600\,\alpha^{18} - 52\,679\,537\,809\,920\,\alpha^{19} - 1\,074\,030\,451\,200\,\alpha^{20}\big)\ \text{Seq}[3+\alpha] + \\
&\big(9\,051\,531\,325\,562\,880 + 90\,332\,029\,095\,081\,984\,\alpha + 420\,333\,410\,362\,428\,416\,\alpha^{2} + \\
&\quad 1\,213\,206\,945\,955\,473\,664\,\alpha^{3} + 2\,437\,377\,188\,874\,087\,136\,\alpha^{4} + 3\,625\,291\,113\,645\,770\,712\,\alpha^{5} + \\
&\quad 4\,144\,688\,219\,837\,114\,384\,\alpha^{6} + 3\,731\,957\,019\,300\,871\,994\,\alpha^{7} + 2\,689\,507\,840\,271\,682\,912\,\alpha^{8} + \\
&\quad 1\,567\,534\,832\,320\,365\,967\,\alpha^{9} + 743\,334\,125\,295\,350\,476\,\alpha^{10} + 287\,455\,002\,784\,035\,524\,\alpha^{11} + \\
&\quad 90\,539\,774\,552\,500\,272\,\alpha^{12} + 23\,112\,095\,925\,472\,389\,\alpha^{13} + 4\,737\,102\,973\,509\,780\,\alpha^{14} + \\
&\quad 767\,930\,664\,461\,310\,\alpha^{15} + 96\,195\,146\,877\,576\,\alpha^{16} + 8\,977\,485\,504\,456\,\alpha^{17} + \\
&\quad 587\,451\,930\,408\,\alpha^{18} + 24\,041\,253\,600\,\alpha^{19} + 462\,944\,160\,\alpha^{20}\big)\ \text{Seq}[4+\alpha]
\end{aligned}
$$